

Tutorial

Calculating the ranking using Excel

Hermann Baum

January 2026

CONTENT

Introduction.....	2
Part 1:	3
Criteria: points score, goal difference and goals scored	3
Part 2	5
Head-to-head record.....	5
2.1 Implementation of head-to-head records in formulas	6
2.2 Decomposition of the function ISNUMBER into its individual parts	7
2.3 Conclusion of the first head-to-head record	8
2.4 Second head-to-head record.....	9
Part 3	10
Further comments on the Excel file	10

Introduction

Determining the ranking within a group usually requires considering several criteria with varying priorities. In the simplest case, these are the number of points, goal difference, and goals scored.

More often, head-to-head results are also taken into account, and these are even used multiple times at the UEFA European Championship and, for the first time, at the 2026 FIFA World Cup. Implementing this in Excel quickly becomes an impenetrable tangle of formulas.

This tutorial will introduce an elegant, relatively streamlined, and very clear way to calculate the ranking within a group.

Head-to-head comparisons, in particular, require a lot of effort—especially when there are more than four teams per group. For example, with nine teams per group, my previous tournament schedules used more than 1,000 array formulas (per group!) on the relevant 'Calc' spreadsheet. The method presented here requires understanding only a single array formula, which is used 27 times with nine teams in a group, or 54 times with repeated head-to-head comparisons.

The tutorial is divided into two parts. The first part discusses how to consider the criteria of points, goal difference, and goals scored. This is accomplished using three simple formulas: one with the RANK function, one with the COUNTIFS function, and a third that simply calculates the sum of two cells.

The second part deals with head-to-head comparisons. It examines the UEFA criteria order for the 2026 FIFA World Cup: points, head-to-head comparison 1 (points, goal difference, goals scored), head-to-head comparison 2 (points, goal difference, goals scored), goal difference across all matches, goals scored across all matches, and fair play.

The head-to-head comparisons use a slightly more complex array formula, which is broken down into its individual components and explained in detail.

A basic introduction to array formulas can be found → [here](#).

An accompanying Excel file is available for the tutorial: → [Download Excel file](#)

Part 1:

Criteria: points score, goal difference and goals scored

	C	D	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X
1	de 2:	FALSE																	
2																			
3			GF	GA	GD	Pts	Pld	W	D	L	Fair play		RANK		GD	RANK		GF	RANK
4	1	1. FC Riedwald	8	7	1	6	5	2	0	3	0		cr.1		add	cr.2		add	cr.3
5	2	FC Barcelona	10	11	-1	6	5	2	0	3	0		2		0	2		0	2
6	3	Eintracht Frankfurt	10	10	0	6	4	2	0	2	0		2		3	5		0	5
7	4	Maibach 1822 eV	3	2	1	4	2	1	1	0	0		2		1	3		0	3
8	5	VFB Essenheim	11	8	3	9	5	3	0	2	0		8		0	8		0	8
9	6	Inter Mailand	3	7	-4	6	4	2	0	2	0		1		0	1		0	1
10	7	SSV Wildbach	2	1	1	4	2	1	1	0	0		2		5	7		0	7
11	8	SV Obersuhl	10	11	-1	6	5	2	0	3	0		8		0	8		1	9
12	9	SSC Bosserode	10	10	0	6	4	2	0	2	0		2		3	5		0	5
													2		1	3		0	3

Fig. 1.1

The **basic idea** in both parts is that a ranking is generated for each criterion, showing the order after that criterion is applied.

Column R shows the ranking after applying the first criterion, points. The numbers in column R indicate the rank of each team. The first team (the team in the top row: 1. FC Riedwald) is in 2nd place after comparing points, the second and third teams are also in 2nd place, the fourth team is in 8th place, and so on.

Team 5 (VFB Essenheim) is in 1st place, leading the table. It is the only team to have achieved 9 points. Six teams share 2nd place, each with 6 points. These rankings in column R are generated by the following simple formula:

=RANG(\$K4,\$K\$4:\$K\$12)

=RANG(\$K5,\$K\$4:\$K\$12)

...

=RANG(\$K12,\$K\$4:\$K\$12)

To apply the second criterion, goal difference, the following formulas are entered in column T:

=COUNTIFS(\$J\$4:\$J\$12,">"&\$J4,\$R\$4:\$R\$12,"="&\$R4)

=COUNTIFS(\$J\$4:\$J\$12,">"&\$J5,\$R\$4:\$R\$12,"="&\$R5)

=COUNTIFS(\$J\$4:\$J\$12,">"&\$J6,\$R\$4:\$R\$12,"="&\$R6)

...

=COUNTIFS(\$J\$4:\$J\$12,">"&\$J12,\$R\$4:\$R\$12,"="&\$R12)

The COUNTIFS function checks two conditions.

First condition: \$J\$4:\$J\$12,">"&\$J4

In each row, it is checked whether the goal difference of the team in that row is better than that of the team in question.

Second condition: \$R\$4:\$R\$12,"="&\$R4

In each row, the formula checks whether the rank in column R is the same as that of the team in question.

In other words, the formula counts how many teams with the same rank have a better goal difference.

For example, if cell T9 contains a "5", this means that five teams with the same rank but a better goal difference were counted for Inter Milan: teams 1, 2, 3, 8, and 9.

The useful thing about this number is that it indicates exactly how many ranks Inter Milan drops due to its inferior goal difference. To their current rank of 2, you simply add the 5 from cell T9. The formula in cell U9 does this.

It reads: =R9+T9

Thus, in accordance with the basic idea mentioned above, a new list of rankings is created, showing the order after applying the criterion 'goal difference'.

The formulas in columns W and X work in the same way, generating the ranking after applying the third criterion (number of goals scored). The formula in column W refers to column H, where the goals scored are listed.

Here, Team 7 (SSV Wildbach) drops to 9th place because Team 4 (Maibach 1822 eV) scored one more goal.

Column X thus shows the ranking after applying the first three criteria. Team 3 (Eintracht Frankfurt) and Team 9 (SSC Bosserode) are still in the same position, namely 3rd place, as are Team 2 (FC Barcelona) and Team 8 (SV Obersuhl), which are both in 5th place.

Head-to-head records would follow here, but in this specific case, they are not relevant because these teams have not yet played against each other.

Part 2

Head-to-head record

In the German Bundesliga and at FIFA World Cups up to 2022, head-to-head records are only used when teams are tied on points, goal difference, and goals scored (Mode 1).

At the UEFA European Championship and, since 2026, also at the FIFA World Cup, head-to-head records are used when teams are tied on points (Mode 2). This increases the likelihood of head-to-head records between more than two teams.

This Excel file implements both modes with formulas: Mode 1 in rows 4 to 12, and Mode 2 in rows 18 to 26. The settings allow you to select which mode is used in the table on the 'Results' sheet.

To examine the formulas for head-to-head records, Mode 2 is the most suitable because it's easier to construct a scenario in which as many teams as possible are subjected to a head-to-head comparison. Therefore, we will examine the formulas in rows 18 to 26.

	C	D	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z	AA	AB	AC	AD	AE	AF	AG
15																												
16				RANK		DC 1		RANK		DC 1		RANK		DC 1		RANK		DC 2		RANK		DC 2		RANK		DC 2		RANK
17				cr.1		Pts	add	cr.2		GD	add	cr.3		GF	add	cr.4		Pts	add	cr.5		GD	add	cr.6		GF	add	cr.7
18	1	1. FC Riedwald		2		6	0	2		1	1	3		8	0	3		0	2	5		-4	0	5		1	0	5
19	2	FC Barcelona		2		6	0	2		3	0	2		10	0	2		0	0	2		0	0	2		0	0	2
20	3	Eintracht Frankfurt		2		6	0	2		0	4	6		10	0	6		0	0	6		0	0	6		0	0	6
21	4	Maibach 1822 eV		8		1	0	8		0	0	8		1	0	8		1	0	8		0	0	8		1	0	8
22	5	VFB Essenheim		1		0	0	1		0	0	1		0	0	1		0	0	1		0	0	1		0	0	1
23	6	Inter Mailand		2		3	5	7		-6	0	7		1	0	7		0	0	7		0	0	7		0	0	7
24	7	SSV Wildbach		8		1	0	8		0	0	8		1	0	8		1	0	8		0	0	8		1	0	8
25	8	SV Obersuhl		2		6	0	2		1	1	3		8	0	3		3	1	4		-1	0	4		2	0	4
26	9	SSC Bosserode		2		6	0	2		1	1	3		8	0	3		6	0	3		5	0	3		5	0	3

Fig. 2.1

In the list of scores **K4:K12** (note: not to be confused with the scores in the K18:K26 range), it can be seen that six teams have achieved a score of 6 (teams 1, 2, 3, 6, 8, and 9). The ranking calculation in the I18:I26 range therefore places all six of these teams in second place.

In Mode 2, a head-to-head comparison of these six teams tied on points is performed. This involves looking at a table that only considers the matches played between these six teams. In this case, the table looks like this:

						FC Barce	1. FC Rie	SV Ober	SSC Bos	Eintrach	Inter Me			
	Participant or team	Pts	GD	GF	bonus									
2.	FC Barcelona	6	3	10	0		0 - 2		7 - 3	3 - 2				
3.	1. FC Riedwald	6	1	8	0	2 - 0		1 - 2	0 - 3	1 - 2	4 - 0			
5.	SV Obersuhl	6	1	8	0		2 - 1		0 - 2	6 - 3	0 - 1			
4.	SSC Bosserode	6	1	8	0	3 - 7	3 - 0	2 - 0						
6.	Eintracht Frankfurt	6	0	10	0	2 - 3	2 - 1	3 - 6			3 - 0			
7.	Inter Mailand	3	-6	1	0		0 - 4	1 - 0		0 - 3				

Fig. 2.2

Looking at the points achieved in the head-to-head comparison, Inter Milan falls to the lowest ranking (7th place).

The remaining five teams are then compared based on their goal difference in a head-to-head record. FC Barcelona finishes in the best position (2nd place) with a goal difference of 3, while Eintracht Frankfurt drops to 6th place.

For the three teams with the same goal difference, the goals scored in the head-to-head record are then examined. However, they are also tied here.

Since these three teams are indistinguishable in the first head-to-head record, they are subjected to a second head-to-head record, in which only the matches between these three teams are examined. The resulting table looks like this:

	Participant or team	Pts	GD	GF	SSC Bos	SV Ober	1. FC Rie							
3.	SSC Bosserode	6	5	5		2 - 0	3 - 0							
4.	SV Obersuhl	3	-1	2	0 - 2		2 - 1							
5.	1. FC Riedwald	0	-4	1	0 - 3	1 - 2								

Fig. 2.3

The points achieved in this second head-to-head record already reveal a clear order of the three teams, so that goal differences and goals scored no longer need to be used for differentiation.

2.1 Implementation of head-to-head records in formulas

As a reminder:

In Part 1, we calculated numbers in the range T4:T12 that were simply added to the previous rankings in the range R4:R12, resulting in the new rankings in the range U4:U12. The COUNTIFS function in cells T4:T12 performed a count in the list of goal differences from the range J4:J12.

The head-to-head comparisons will work exactly the same way. The range L18:L26 again uses this COUNTIFS function, whose results are added to the previous rankings in the range I18:I26, resulting in the new rankings in the range M18:M26.

However, this time the COUNTIFS function isn't counting in a list of goal differences, but in a list that doesn't yet exist and therefore has to be created. And that brings us to the main problem and the most difficult step of this tutorial: the announced array formula.

We need a list of points earned. For each team, we need to determine how many points it has earned in matches against the teams with the same number of points. For the six teams tied on points, we need to calculate the same numbers as shown in the "Pts" column of Figure 2.2.

This task is performed by the array formulas in cells K18:K26.

In cell K18 there is the following array formula:

```
{=SUMPRODUCT($O$31:$O$102*($J$31:$J$102=$D18)*ISNUMBER(MATCH($K$31:$K$102,IF($I$18:$I$26=$I18,$D$18:$D$26,""),0)))+SUMPRODUCT($P$31:$P$102*($K$31:$K$102=$D18)*ISNUMBER(MATCH($J$31:$J$102,IF($I$18:$I$26=$I18,$D$18:$D$26,""),0)))}
```

It consists of two identical addends:

```
{=SUMPRODUCT(...) + SUMPRODUCT(...)}
```

We will first consider the first summand:

```
SUMPRODUCT($O$31:$O$102*($J$31:$J$102=$D18)*ISNUMBER(MATCH($K$31:$K$102,IF($I$18:$I$26=$I18,$D$18:$D$26,""),0)))
```

The first part, without the ISNUMBER function, is:

```
=SUMPRODUCT($O$31:$O$102*($J$31:$J$102=$D18)).
```

It adds up all scores in the range O31:O102 where the name "1. FC Riedwald" appears in column J on the left - i.e., all points achieved by 1. FC Riedwald in home games.

To limit the total points to those scored against the five teams with the same number of points, the parameters within the SUMPRODUCT function must be multiplied by another factor.

This third factor must be a list of TRUE/FALSE values: TRUE if the opponent is a team with the same number of points, FALSE otherwise. This list of TRUE/FALSE values is generated by the ISNUMBER function.

2.2 Decomposition of the function ISNUMBER into its individual parts

This expression will now be considered:

```
ISNUMBER(MATCH($K$31:$K$102, IF($I$18:$I$26=$I18, $D$18:$D$26, ""), 0)).
```

We'll proceed from the inside out. First, the IF function:

```
IF($I$18:$I$26=$I18, $D$18:$D$26, "")
```

Within the IF function, the condition `I18:I26=$I18` generates not just a single Boolean value (TRUE or FALSE), but an entire list (array) of Boolean values. You can view this list (array) by selecting the range AR31:AR39 on the 'Calc' worksheet, entering the formula `= I18:I26=$I18` in the formula bar at the top, and pressing Ctrl+Shift+Enter.

In newer versions of Excel, from Excel 2021 and Excel 365 onwards, you can also simply enter this (array) formula into cell V31 and press Enter. The system will automatically recognize that it is an array formula.

The IF function uses this list of Boolean values to filter out only those teams from the list that meet the condition, i.e., those that have the same rank in column I as 1. FC Riedwald. To display the result of the IF function, select the range AS31:AS39, enter the (array) formula

```
=IF($I$18:$I$26=$I18, $D$18:$D$26, ""),
```

and press Ctrl+Shift+Enter.

The range AS31:AS39 will now display the following:

TRUE	1. FC Riedwald
TRUE	FC Barcelona
TRUE	Eintracht Frankfurt
FALSE	
FALSE	
TRUE	Inter Mailand
FALSE	
TRUE	SV Obersuhl
TRUE	SSC Bosserode

Fig. 2.4

The SUMPRODUCT function should add all points in the range O31:O102 if column J contains the team "1. FC Riedwald" and column K contains a team from the list of teams with the same number of points (see Fig. 2.4). Therefore, we need another list of 72 TRUE/FALSE values, which contains a TRUE value for each team in the range K31:K102 if it is one of the six teams tied on points, and a FALSE value otherwise.

To achieve this, the MATCH function determines the index in the list of the six teams for each team in the range K31:K102. The MATCH function generates the #N/A error for any team that is not in the list of the six teams tied on points.

We can also visualize the list of indices generated by the MATCH function: Select the range AT31:AT102, enter the (array) formula `=MATCH(K31:K102, IF(I18:I26=$I18, D18:D26, ""), 0)` in the input field above, and complete the entry with CTRL+SHIFT+ENTER (see Figure 2.5 for results).

The ISNUMBER function converts any #N/A error code to FALSE and a valid index to TRUE.

We can make the list of these 72 TRUE/FALSE values visible by selecting the range AU31:AU102, entering the (array) formula `=ISNUMBER(MATCH(K31:K102, IF(I18:I26=$I18, D18:D26, ""), 0))` into the formula bar above and completing the entry with CTRL-SHIFT-ENTER.

TRUE	1. FC Riedwald	2	TRUE
TRUE	FC Barcelona	8	TRUE
TRUE	Eintracht Frankfurt	#N/A	FALSE
FALSE		6	TRUE
FALSE		1	TRUE
TRUE	Inter Mailand	#N/A	FALSE
FALSE		3	TRUE
TRUE	SV Obersuhl	#N/A	FALSE
TRUE	SSC Bosserode	8	TRUE
		9	TRUE
		2	TRUE
		#N/A	FALSE
		1	TRUE
		8	TRUE
		#N/A	FALSE
		3	TRUE
		6	TRUE
		#N/A	FALSE
		#N/A	FALSE
		9	TRUE
		1	TRUE

Fig. 2.5

Note:

The four lists in Figure 2.5 are generated internally (invisibly) by Excel when evaluating the array formulas and are processed immediately, without needing to be stored as intermediate results on the worksheet. This improves performance.

However, there is a second aspect to consider when using array formulas: The four lists shown here only apply to the first SUMPRODUCT function of the array formula in cell K18. For the second SUMPRODUCT function and for the array formulas in other cells, the four lists may look completely different. For the 81 array formulas on the 'Calc' worksheet, $81 \times 8 = 648$ of these lists are needed. Displaying them all as intermediate results would create a large and confusing jumble of formulas.

2.3 Conclusion of the first head-to-head record

In the formula `{=SUMPRODUCT(...)+SUMPRODUCT(...)}`, there are two similarly structured addends. The first adds all the points the team in question scored as the home team against teams with the same number of points. The second addend adds all the points the team in question scored as the away team against teams with the same number of points.

In the range K18:K26 we have the list of points each of the six teams tied on points scored against the others (see also Figure 2.2, column "Points").

The COUNTIFS function in the range L18:L26 now counts for each team how many teams tied on points scored higher in the head-to-head record. For the team "Inter Milan," there is a 5 because Inter Milan only scored 3 points in the head-to-head record, while the other 5 teams each scored 6 points. Therefore, Inter Milan falls back to 7th place.

For the other five teams, the next step involves checking the goal differences from the head-to-head record.

In the O18:O26 range, the same array formula is used as in the K18:K26 range, but now it calculates the goal differences of the six teams tied on points. The formulas calculate the same numbers as shown in the "Diff." column of Figure 2.2 above.

The formulas in the P18:P26 range now count, for each team, how many of the teams still tied have a better goal difference. Of the five teams still in second place, three drop to third and one to sixth. FC Barcelona, with the best goal difference of +3, remains the only team in second place.

In the next step (formulas in the S18:U26 range), the goals scored by those teams that are still in the same rank are checked using the same mechanism (ranking in the Q18:Q26 range).

However, the three teams in third place do not differ in terms of goals scored. Therefore, their rankings remain unchanged.

2.4 Second head-to-head record

The second head-to-head record proceeds according to the same pattern. The crucial difference is that the array formulas of the first head-to-head record all refer to the rank numbers in the range I18:I26, while the array formulas of the second head-to-head record refer to the rank numbers in the range U18:U26.

The array formulas of the second head-to-head record calculate the numbers in Figure 2.3, columns "Points," "Diff.," and "Goals Scored."

The final rankings of the remaining three teams are determined in the very first step of this second head-to-head record, by comparing the points achieved. The comparison of goal difference and goals scored then has no further effect.

Part 3

Further comments on the Excel file

After the second head-to-head comparison, the following criteria are used:

- Goal difference from all group matches
- Goals scored from all group matches
- Tie-break (fair play, world ranking, or draw).

The final rankings are displayed in the range E4:E12 for Mode 1 and in the range E18:E26 for Mode 2. The selection on the 'Settings' worksheet determines which of these two lists appears in the range A4:A12.

The formulas in the table on the 'Results' worksheet retrieve the current rankings from this range A4:A12, so that the table on the 'Results' worksheet displays the ranking order according to either Mode 1 or Mode 2, depending on the settings selection.

